



基于 HSPICE 的集成电路仿真

HSPICE 仿真器是 Synopsys 公司的电路级仿真工具,采用和标准 SPICE 描述一致的输入文件。在兼容标准 SPICE 的基础上,还提供更强大的集成电路仿真分析能力。本章将介绍与 HSPICE 仿真器相关的流程、工具的使用以及基本电路分析的方法。

3.1 流程及规则简介

HSPICE 工具的基本仿真流程如图 3-1 所示。启动后, HSPICE 工具会运行一些脚本文件、检查 license、读取配置文件步骤,这几个步骤用户一般不需要干预。此时用户需要输入网表文件,如果网表文件中包括库文件引用,则 HSPICE 工具同时会读入库文件。然后 HSPICE 工具分析电路的工作点,进入用户需要的分析仿真。仿真完毕,输出结果并清理文件。HSPICE 输入网表文件通常采用 .sp 文件,输出文件有 .st0 运行状态文件、.lis 输出列表文件、.tr 瞬态分析文件、.sw 直流分析文件、.ac 交流分析文件等。.st0 运行状态文件和 .lis 输出列表文件内容在每次 HSPICE 运行后均有出现,其他的输出文件根据 SPICE 程序中选择的分析类型而出现,并且可以在波形显示工具中显示,如 Avancees、Cscope 等波形显示工具。

SPICE 网表(程序)输入文件和库输入文件可以由一个电路网表转换器或用一个文本编辑器产生。值得一提的是, HSPICE 等 EDA 工具一般都是基于 Linux 或 UNIX 平台的,因此注意网表输入文件的扩展名并不是强制的。SPICE 网表输入文件一般遵循与标准 SPICE 相同的规则,并且提供了强大的功能以及一些具有自身特色的选项。一些基本注意事项如下。



图 3-1 HSPICE 工具的基本仿真流程

(1) 网表输入文件的第 1 条语句必须是标题行,最后一条语句必须是 .end 语句,它们之间的语句次序是随意的,除非是续行(行首有+符号的行)必须接在要接下去的行后面。注释行以 * 开头,可加在文件中的任何地方。

(2) HSPICE 采用自由格式输入。语句中的域由一个或多个空格、一个 Tab、一个逗号、一个等号或一个左/右圆括号分开。

(3) 除 UNIX、Linux 系统中的文件名外,不区分大小写字母。

(4) 每行语句长度一般限于 80 个字符以下。

(5) 一条语句若在一行写不下,可以用续行号继续下去。续行以+作为第 1 个非数值、非空格字符。

(6) 网表输入文件不能被“打包”,也不能被压缩。

(7) 网表输入文件中不要采用特殊的控制字符。

(8) 电路信息以及波形输出需要在 option 中增加如下声明。

```
.option list node post
```

其中,list 选项表示打印网表元器件、节点连接以及元器件值、电压源和电流源以及参数等; node 选项表示 HSPICE 将打印节点交叉引用对照表,对照表列举每个节点以及连接节点的元器件; post 选项表示保存仿真结果以便由交互的波形查看工具显示仿真结果。

(9) 模型文件可以采用 .lib 进行引用。例如:

```
.lib './library/mix025.lib' TT
```

在这个例子中,mix025.lib 是选取的某个集成电路工艺的 SPICE 模型文件,引用时需要注意带上路径; TT 是选择的模型文件中的 section,section 用于表明模型的工艺角(Corner)。

3.2 HSPICE 工具的使用

HSPICE 可采用命令行或图形界面的方式执行。在 UNIX 或 Linux 系统命令提示符(\$)下执行^①

```
$ hspice demo.sp
```

其中,demo.sp 是输入文件,在这种情况下不生成 .lis 文件,报表文件的内容打印到屏幕上。

```
$ hspice -i demo.sp -o demo.lis
```

在这种情况下,将报表生成以输出文件名 demo 命名的 .lis 文件。在 Windows 操作系统中,启动 HSPICE 相对方便的方式是采用图形界面,如图 3-2 所示,在 Design 文本框中输入 SPICE 输入文件。

单击 Simulate 按钮执行仿真,之后采用 AvanWaves 或 Cscope(CosmosScope)软件显示波形,分别如图 3-3 和图 3-4 所示,双击相应节点就可以显示波形。在 HSPICE 工具各个版本中,图形界

^① \$ 是 UNIX 或 Linux 的命令提示符。




图 3-2 HSPICE 仿真图形界面




图 3-3 AvanWaves 波形查看软件界面




图 3-4 Cscope 波形查看软件界面

面会有一些差异,但操作方式是大同小异的。Cscope(CosmosScope)的显示功能要比 AvanWaves 丰富些。在 HSPICE 的后续版本中, Linux 平台不再支持 AvanWaves。

3.3 HSPICE 基本电路分析

下面以如图 3-5 所示的电路为例,说明 HSPICE 的几种基础仿真: 直流仿真、交流仿真、瞬态仿真。

此电路为电流源作负载的共源极放大器,采用电流镜实现电流源,偏置为电阻与电流镜实现的简单偏置电路,放大器的输出驱动一个 5pF 的电容负载。电路连接的各节点号(节点 0 到节点 4)已标注在图中,其中电路的地节点(gnd)连接到默认节点 0 的大地节点(ground)。放大器中所有晶体管的尺寸为 $W=5\mu\text{m}$, $L=1\mu\text{m}$; NMOS 晶体管的参数为: $V_{\text{TH}}=0.7\text{V}$, $K_n=110\mu\text{A}/\text{V}^2$, $\lambda=0.04\text{V}^{-1}$, $\gamma=0.4$; PMOS 晶体管的参数为: $V_{\text{TH}}=0.7\text{V}$, $K_p=50\mu\text{A}/\text{V}^2$, $\lambda=0.05\text{V}^{-1}$, $\gamma=0.6$; 电阻的阻值为 $100\text{k}\Omega$ 。




图 3-5 一个基本的共源极放大器

3.3.1 直流仿真分析

针对图 3-5 电路,采用子电路描述的方式,将放大器 AMP 作为一个子电路进行描述。

图 3-5 电路的直流仿真 SPICE 描述如下。

```
* DC analysis for AMP
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 5.0

.OP
.DC Vin 0 5 0.1
.print dc V(2)
.plot dc V(2)
* .probe dc V(2)
.option list node post
```

```

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

MOSFET 器件模型采用 MOS1 模型。放大器子电路名称为 AMP,其端口名为 out、in、vdd 和 gnd。在例化时,例化名为 X1,对应的端口的连线分别为节点 2、节点 1、节点 4 和节点 0。

.OP 是分析直流工作点语句。该语句在进行电路直流工作点计算时,电路中所有电感短路,电容开路。值得注意的是,在一个 HSPICE 模拟中只能出现一个 .OP 语句。

采用 HSPICE 执行仿真,将产生直流工作点分析(.OP)和直流扫描分析(.DC)结果。HSPICE 首先会统计元器件模型参数以及电路中诸如 MOS 晶体管、电阻、电容、独立源等各个元器件的参数,然后分析电路中各个元器件的工作点。这些都会打印到屏幕上或在 .lis 文件中显示出来。

关于电路中各 MOS 晶体管工作点的部分报表如图 3-6 所示。从中可以得知诸如漏极电流(id)、各端口之间的偏压等工作点信息,进而可以得知 gm、gds、gmb 等小信号参数数值。还可以发现,M2 和 M3 处于饱和区,而 M1 处于线性区。为什么 M1 处于线性区而不是饱和区呢?这是因为在上述 SPICE 描述中,输入处(节点 1)的独立源 V_{in} 施加的直流电压值是 5V,即 M1 的 $V_{GS}=5V$,这样使 M1 进入了线性区。结合后续的直流扫描,可以将 V_{in} 直流值设置在 1~1.12V,如 1.07V。重新进行 HSPICE 仿真,晶体管工作点的部分报表如图 3-7 所示,可见所有晶体管都处于饱和区,这是电路设计所期望的。这样可以通过 HSPICE 仿真的 .OP 报表结果,分析电路的直流工作点,调整电路设计以便符合设计目标。

```

**** mosfets
subckt x1      x1      x1
element 1:m1  1:m2  1:m3
model 0:mosn  0:mosp  0:mosp
region      Linear Saturati Saturati
id          44.3226u -44.3226u -37.6720u
ibs        0.        0.        0.
ibd        -187.6795a 49.8123f  12.3280f
vgs        5.0000   -1.2328  -1.2328
vds        18.7680m -4.9812  -1.2328
vbs        0.        0.        0.
vth        700.0000m -700.0000m -700.0000m
vdsat      18.7680m -532.8017m -532.8017m
vod        4.3000   -532.8017m -532.8017m
beta       550.4129u 312.2654u  265.4100u
gam eff    400.0000m 600.0000m 600.0000m
gm         10.3301u  166.3755u  141.4109u
gds        2.3582m  1.7742u   1.7742u
gmb        2.4694u  55.8040u  47.4307u
cdtot      862.0302a  11.4673a  2.8380a
cgtot      1.7281f  1.1782f  1.1696f
cstot      864.5476a 1.1511f  1.1511f
cbtot      1.5421a  15.6894a  15.6894a
cgs        864.5476a 1.1511f  1.1511f
cgd        862.0302a 11.4673a  2.8380a

```

```

**** mosfets
subckt x1      x1      x1
element 1:m1  1:m2  1:m3
model 0:mosn  0:mosp  0:mosp
region      Saturati Saturati Saturati
id          40.7273u -40.7273u -37.6720u
ibs        0.        0.        0.
ibd        -20.4515f 29.5485f  12.3280f
vgs        1.0700   -1.2328  -1.2328
vds        2.0452  -2.9548  -1.2328
vbs        0.        0.        0.
vth        700.0000m -700.0000m -700.0000m
vdsat      370.0000m -532.8017m -532.8017m
vod        370.0000m -532.8017m -532.8017m
beta       594.9934u 286.9356u  265.4100u
gam eff    400.0000m 600.0000m 600.0000m
gm         220.1475u  152.8798u  141.4109u
gds        1.5059u   1.7742u   1.7742u
gmb        52.6253u  51.2774u  47.4307u
cdtot      4.7082a  6.8024a  2.8380a
cgtot      1.1728f  1.1735f  1.1696f
cstot      1.1511f  1.1511f  1.1511f
cbtot      17.0822a  15.6894a  15.6894a
cgs        1.1511f  1.1511f  1.1511f
cgd        4.7082a  6.8024a  2.8380a

```

图 3-6 静态工作点分析得到的电路中 MOSFET 工作点情况 ($V_{in}=5V$)

图 3-7 $V_{in}=1.07V$ 时的 MOSFET 工作点情况

直流分析针对 V_{in} 从 0V 变化到 5V, 步长为 0.1V。通过直流扫描分析, 可以得到此放大器的输入/输出特性。执行 .print dc V(2) 语句后, 会在 .lis 文件或屏幕上打印输出节点 2 的电压随 V_{in} 直流扫描电压变化的数据, 如图 3-8 所示。图 3-9 所示为执行 .plot dc V(2) 语句得到的采用文本方式显示的图 3-5 电路的直流扫描结果。

volt	v(2)
(a)	0.
0.	5.000
100.0000m	5.000
200.0000m	5.000
300.0000m	5.000
400.0000m	5.000
500.0000m	5.000
600.0000m	5.000
700.0000m	5.000
800.0000m	4.975
900.0000m	4.891
1.0000	4.696
1.1000	344.415m
1.2000	197.941m
1.3000	152.054m
1.4000	125.270m
1.5000	107.120m
1.6000	93.829m
1.7000	83.604m
1.8000	75.463m
1.9000	68.810m
2.0000	63.262m
2.1000	58.561m
2.2000	54.522m
2.3000	51.012m
2.4000	47.934m
2.5000	45.210m
2.6000	42.782m
2.7000	40.605m
2.8000	38.640m
2.9000	36.859m
3.0000	35.235m
3.1000	33.750m
3.2000	32.385m
3.3000	31.127m
3.4000	29.964m
3.5000	28.885m
3.6000	27.881m
3.7000	26.945m
3.8000	26.071m
3.9000	25.251m
4.0000	24.481m
4.1000	23.758m
4.2000	23.076m
4.3000	22.432m
4.4000	21.823m
4.5000	21.246m
4.6000	20.699m
4.7000	20.180m
4.8000	19.686m
4.9000	19.216m
5.0000	18.768m

图 3-8 节点 2 的电压随 V_{in} 直流扫描电压变化的数据(部分)

volt	v(2)	2.0000	4.0000	6.0000	8.0000
(a)	0.				
0.	5.000	+	+	+	+
100.0000m	5.000	+	+	a	+
200.0000m	5.000	+	+	a	+
300.0000m	5.000	+	+	a	+
400.0000m	5.000	+	+	a	+
500.0000m	5.000	+	+	a	+
600.0000m	5.000	+	+	a	+
700.0000m	5.000	+	+	a	+
800.0000m	4.975	+	+	a	+
900.0000m	4.891	+	+	a+	+
1.0000	4.696	+	+	a-	+
1.1000	344.415m	+	+	+	+
1.2000	197.941m	+	+	+	+
1.3000	152.054m	+	+	+	+
1.4000	125.270m	+	+	+	+
1.5000	107.120m	+	+	+	+
1.6000	93.829m	+	+	+	+
1.7000	83.604m	+	+	+	+
1.8000	75.463m	+	+	+	+
1.9000	68.810m	+	+	+	+
2.0000	63.262m	+	+	+	+
2.1000	58.561m	+	+	+	+
2.2000	54.522m	+	+	+	+
2.3000	51.012m	+	+	+	+
2.4000	47.934m	+	+	+	+
2.5000	45.210m	+	+	+	+
2.6000	42.782m	+	+	+	+
2.7000	40.605m	+	+	+	+
2.8000	38.640m	+	+	+	+
2.9000	36.859m	+	+	+	+
3.0000	35.235m	+	+	+	+
3.1000	33.750m	+	+	+	+
3.2000	32.385m	+	+	+	+
3.3000	31.127m	+	+	+	+
3.4000	29.964m	+	+	+	+
3.5000	28.885m	+	+	+	+
3.6000	27.881m	+	+	+	+
3.7000	26.945m	+	+	+	+
3.8000	26.071m	+	+	+	+
3.9000	25.251m	+	+	+	+
4.0000	24.481m	+	+	+	+
4.1000	23.758m	+	+	+	+
4.2000	23.076m	+	+	+	+
4.3000	22.432m	+	+	+	+
4.4000	21.823m	+	+	+	+
4.5000	21.246m	+	+	+	+
4.6000	20.699m	+	+	+	+
4.7000	20.180m	+	+	+	+
4.8000	19.686m	+	+	+	+
4.9000	19.216m	+	+	+	+
5.0000	18.768m	+	+	+	+

图 3-9 采用文本方式显示的直流扫描结果

为了绘制曲线波形图, SPICE 需要执行 PROBE 命令, 而在 HSPICE 中, 为了得到更好的曲线波形显示效果, 需要增加 .option list node post 控制语句, 执行直流扫描后会产生 .sw(或 .sw0) 文件, 就可以配合使用一些图形后处理工具绘制波形图。图 3-10 所示为采用 AvanWaves 打开 .sw(或 .sw0) 文件查看的直流扫描结果。

采用 Cscope(CosmosScope) 打开结果文件, 如图 3-11 所示。在 Cscope 中执行 File→Open→Plotfiles 菜单命令, 弹出 Open Plotfiles 对话框, 选择 HSPICE(*.tr*, *.ac*, *.sw*, *.ft*) 文件类型, 打开 .sw(或 .sw0) 文件查看直流扫描结果, 如图 3-11(a) 所示。在 Signal Manager 对话框

的子栏中双击 v(2), 就可以显示节点 2 的电压随直流扫描电压的变化曲线。执行 Graph→Color Map 菜单命令, 或者在 Cscope 波形图上右击, 在弹出的快捷菜单中选择 Color Map, 选择 Map 2 波形配色方案, 可以将背景改为白色, 如图 3-11(b) 所示。同样, 也可以通过右击波形曲线, 在弹出的快捷菜单中选择 Color, 改变波形曲线颜色, 同时还可以增加 Measurement 光标, 显示效果如图 3-11(c) 所示。






图 3-10 采用 AvanWaves 查看直流扫描结果




(a) 采用 Cscope 打开 .sw0 文件

图 3-11 采用 Cscope 查看直流扫描结果



(b) 改变波形显示配色方案



(c) 直流扫描显示效果

图 3-11 (续)

从这些不同形式的输出结果中可以看出,输入为 1~1.12V 的区域是此放大器的高增益区。这里详细展示了几种输出控制形式以及波形显示工具,它们对于输出数据的显示与处理各有千

秋,视需要选用。后面采用 Cscope 作为主要的输出波形工具,其他形式的输出结果就不一一赘述了。

【例 3-1】 采用某工艺厂家提供的 $0.25\mu\text{m}$ CMOS 工艺模型文件(mix025.lib)对图 3-5 放大器进行 HSPICE 仿真。输入 V_{GS} 从 0V 变化到 5V ,考查输入/输出特性。

解 前面讲述的 SPICE 描述采用 MOS1 的 MOSFET 模型。在集成电路设计过程中,需要根据工艺厂家提供的器件模型开展电路设计。这样需要在 SPICE 仿真中进行工艺库文件的定义,在 HSPICE 中,采用以下语句进行工艺库文件定义。

```
.LIB 'libfilename' section
```

其中,section 是工艺库的工艺角。工艺角用于说明集成电路制造工艺中出现的偏差。例如,对于 MOS 器件可能出现的工艺角为:典型(typical); NMOS 是快速晶体管且 PMOS 是快速晶体管(fast NMOS,fast PMOS); NMOS 是慢速晶体管且 PMOS 是慢速晶体管(slow NMOS,slow PMOS); NMOS 是快速晶体管且 PMOS 是慢速晶体管(fast NMOS,slow PMOS); NMOS 是慢速晶体管且 PMOS 是快速晶体管(slow NMOS,fast PMOS)。section 的名称没有强制规定,具体名称需要查阅特定工艺厂家的模型文件说明。

电路的 SPICE 描述如下。

```
* DC analysis for AMP using 0.25um CMOS process
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd nch w=5u l=1.0u
M2 out 3 vdd vdd pch w=5u l=1.0u
M3 3 3 vdd vdd pch w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 5.0

.OP
.DC Vin 0 5 0.1
.plot dc V(2)
.probe DC V(2)
.option list node post

.lib './mix025.lib' TT

.end
```

mix025.lib 文件要带路径进行说明。这里和仿真用的 SPICE 描述文件处于同一路径下。TT 是工艺的典型工艺角名称,表示典型的 Corner。NMOS 和 PMOS 器件模型名称要查阅 mix025.lib 工艺文件及说明,这里分别为 nch 和 pch。执行 HSPICE 仿真后,使用 Cscope 打开.sw0 文件,得到采用此 $0.25\mu\text{m}$ 工艺放大器的输入/输出特性曲线,如图 3-12 所示。可以看出,采用不同的工艺,

放大器的输入/输出特性发生了变化,为了使放大器处于高增益区,输入可偏置在 0.75V。




图 3-12 在某 0.25 μm CMOS 工艺下的直流扫描 HSPICE 仿真输入/输出特性曲线

3.3.2 交流仿真分析

根据直流分析的结果,将图 3-5 放大器电路的输入直流工作点 V_{in} 设置为 1.07V,然后对电路进行交流小信号分析。针对图 3-5 电路的交流仿真 SPICE 描述如下。

```
* AC analysis for AMP
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 1.07 AC 1.0

.OP
.AC DEC 20 100 100MEG
.plot ac VDB(2) VP(2)
.probe
.option list node post

* model
```

```

.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

. AC 语句根据计算的直流工作点,将电路中的非线性元器件(如 MOSFET)采用小信号模型进行等效,然后分析电路的频率特性,如幅频特性、相频特性。在本例中,. AC DEC 20 100 100MEG 语句表示 100Hz~100MHz 进行交流扫描分析,每 10 倍频程扫描 20 个点。为了能够得到交流仿真的结果,在放大器的输入端施加的输入信号必须包含交流输入成分,如 Vin 1 0 DC 1.07 AC 1.0 表示在节点 1 和节点 0 之间施加的输入电压源 V_{in} 的直流偏置为 1.07V,而交流输入为 1V(单位 1)。

采用 HSPICE 执行仿真,将产生交流小信号分析(. AC)结果文件。采用 Cscope 打开 .ac0 结果文件。此处 HSPICE 描述打印输出节点(节点 2)的幅频特性曲线(上)和相频特性曲线(下),幅频特性曲线中输出以分贝(dB)为单位,相频特性曲线的输出单位为度($^{\circ}$),如图 3-13 所示。可见随着频率的增加,在高于 3dB 频率以上,幅频特性以每 10 倍频程 20dB 下降,而相频特性从 180° (反相)开始发生相移,最大相移约为 90° 。放大器的 3dB 带宽约为 104kHz。




图 3-13 HSPICE 交流仿真结果

3.3.3 瞬态仿真分析

为了了解在施加瞬态输入激励(如正弦信号)的情况下电路随时间变化的瞬态行为,对图 3-5 电路进行瞬态仿真。这里,首先考查在输入端施加偏置为 2.0V、振幅为 1.0V、频率为 100kHz 的

正弦信号时的电路行为。瞬态仿真 SPICE 描述如下。

```

* TRAN analysis for AMP
*
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 2.0 sin(2.0 1.0 100K)

.OP
.TRAN 0.1u 30u
.plot tran v(2) v(1)
.probe
.option list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

在本例中, `.TRAN 0.1u 30u` 语句表示瞬态仿真的结束时间为 $30\mu\text{s}$, 步长为 $0.1\mu\text{s}$, 起始时间为 0 时刻。同样, 为了能够得到瞬态仿真结果, 在放大器的输入端施加的输入信号必须包含瞬态输入成分, 如 `Vin 1 0 DC 2.0 sin(2.0 1.0 100K)`。

采用 HSPICE 执行仿真, 将产生瞬态分析 (`.TRAN`) 结果文件。采用 Cscope 打开 `.tr0` 结果文件。在本例中, 偏置为 2.0V , 振幅为 1.0V , 频率为 100kHz 的正弦信号输入时的瞬态仿真结果如图 3-14 所示。可见当输入信号 `v(1)` 幅度很大时, 放大器达到了饱和, 因而在输出端输出信号 `v(2)` 不能得到输入的放大信号。

当输入调整为小信号时, 并注意偏置值的选取, 根据直流仿真的结果, 选择 1.07V 的偏置, 正弦信号的幅度为 0.0001V , 即 `Vin 1 0 DC 1.07 sin(1.07 0.0001 100K)`, 得到的瞬态仿真结果如图 3-15 所示。

通过瞬态仿真, 可见此时放大器处于正确的偏置下, 输入偏置为 1.07V , 输出偏置约为 2.045V , 这与采用直流扫描得到的输入/输出特性曲线是可以对应得上。在此偏置下, 可以对输入正弦信号进行放大。由仿真波形图可知, 小信号增益约为 50 倍 (34dB)。与图 3-13 的交流仿真结果进行对照, 可以发现在 100kHz 时的增益结果是一致的, 同样, 相位的结果也是一致的。

【例 3-2】 进行 NMOS 放大器的 HSPICE 仿真。如图 3-16 所示的 NMOS 放大器, $V_{\text{GS}}=1\text{V}$,




图 3-14 HSPICE 瞬态仿真结果(偏置为 2V,幅度为 1V)




图 3-15 HSPICE 瞬态仿真结果(偏置为 1.07V,幅度为 0.0001V)




图 3-16 电阻作负载的 NMOS 共源极放大器

$V_{DD}=5\text{V}$, $R_D=10\text{k}\Omega$ 。已知 NMOS 放大器的参数为: $W=5\mu\text{m}$, $L=1\mu\text{m}$, $V_{TH}=0.7\text{V}$, $K_n=110\mu\text{A}/\text{V}^2$, $\lambda=0.04\text{V}^{-1}$ 。进行 SPICE 仿真: 输入 V_{GS} 从 0V 变化到 5V , 考查输入/输出特性; 在 $V_{GS}=1.0\text{V}$ 的偏置下, 并且在驱动 $C_{LD}=1.0\text{pF}$ 的电容负载情况下, 考查电路的幅频特性与相频特性; 在 $V_{GS}=1.0\text{V}$ 的偏置下, 叠加幅度为 0.01V , 频率为 100kHz 的正弦信号, 考查输出的瞬态特性。

解 电路的 SPICE 描述如下。

```
* DC, AC and TRAN analysis for AMP in chapter 3
M1 out in gnd gnd MOSN w=5u l=1.0u
RD out vdd 10K
CLD out gnd 1p

Vdd vdd 0 DC 5.0
Vgnd gnd 0 DC 0.0
Vgs in 0 DC 1.0 AC 1.0 sin(1.0 0.01 100KHZ)

.OP
.DC Vgs 0 5 0.1
.AC DEC 20 100 1000MEG
.TRAN 0.1u 30u

.save all
.plot dc V(out)
.plot dc I(Vdd)
.plot ac V(out)
.plot ac VP(out)
.plot tran V(in)
.plot tran V(out)
.probe
.OPTION list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.end
```

电路的静态工作点报表如图 3-17 所示,可见当输入 V_{GS} 为 1.0V 时,输出约为 4.7059V。

计算一下当输入 V_{GS} 为 1.0V 时输出的电压值。希望将 NMOS 晶体管偏置在其饱和区,因此,根据 MOS 晶体管饱和区漏极电流公式,并忽略沟道长度调制效应, $V_{GS}=1V, V_{TH}=0.7V$, 有

$$\begin{aligned} I_D &= \frac{1}{2} K_n \frac{W}{L} (V_{GS} - V_{TH})^2 = \frac{1}{2} \times 110 \times 10^{-6} \times \frac{5 \times 10^{-6}}{1 \times 10^{-6}} \times (1 - 0.7)^2 \\ &= 24.75 \times 10^{-6} \text{ A} = 24.75 \mu\text{A} \end{aligned}$$

则

$$V_{DS} = V_{DD} - R_D I_D = 5 - 10 \times 10^3 \times 24.75 \times 10^{-6} = 4.7525 \text{ V}$$

这与仿真结果 4.7059V 很接近。

直流扫描分析的 HSPICE 仿真结果如图 3-18 和图 3-19 所示。图 3-18 是放大器的电压转移特性仿真结果,可以看到高增益区在 0.7~1.8V 的输入范围内。

图 3-19 是电源电流 I_{DD} 随输入变化的仿真结果, I_{DD} 与流经 NMOS 晶体管的电流大小相等,方向相反。可见当输入 $V_{GS}=1.0V$ 时,支路电流与计算结果相仿。

交流分析的 HSPICE 仿真结果如图 3-20 所示。

```

***** operating point information      tnom= 25.000 temp= 25.000      **** mosfets
*****
***** operating point status is all      simulation time is 0.
node      =voltage      node      =voltage      node      =voltage
+0:in      = 1.0000  0:out      = 4.7059  0:vdd      = 5.0000

**** voltage sources

subckt
element 0:vdd      0:vgs
volts    5.0000    1.0000
current  -29.4089u  0.
power    147.0443u  0.

total voltage source power dissipation= 147.0443u      watts

**** resistors

subckt
element 0:rd
r value  10.0000k
v drop  -294.0886m
current  -29.4089u
power    8.6488u

subckt
element 0:m1
model    0:mosn
region   Saturati
id       29.4089u
ibs      0.
ibd      -47.0591f
vgs      1.0000
vds      4.7059
vbs      0.
vth      700.0000m
vdsat    300.0000m
vod       300.0000m
beta     653.5301u
gam eff  400.0000m
gm       196.0590u
gds      990.0000n
gmb      46.8671u
cdtot    10.8335a
cgtot    1.1827f
cstot    1.1511f
cbtot    20.8189a
cgs      1.1511f
cgd      10.8335a

```

(a) 节点电压、独立源及电阻工作点

(b) MOS晶体管工作点

图 3-17 电路的静态工作点报表




图 3-18 放大器的电压转移特性 HSPICE 仿真结果

当 NMOS 晶体管偏置在饱和区时, NMOS 放大器的增益可以表示为

$$A_v = -g_m (R_D \parallel r_o)$$

从图 3-17 的工作点报表中可知, 晶体管 M1 的 $g_m \approx 196.06 \mu\text{A}/\text{V}$, $r_o = 1/g_{ds} = (1/990\text{n}) \Omega$, $R_D = 10\text{k}\Omega$, 由此计算得到直流增益 $A_v \approx 1.95 \approx 5.8\text{dB}$ 。可见仿真结果和计算结果一致。

还可以计算一下放大器的 3dB 带宽, 从图 3-17 的工作点报表中可知, 晶体管 M1 的 $r_o = 1/g_{ds} = (1/990\text{n}) \Omega$, $R_D = 10\text{k}\Omega$, 那么 3dB 带宽为



图 3-19 电源电流 I_{DD} 随输入变化的 HSPICE 仿真结果

图 3-20 放大器的幅频特性和相频特性的 HSPICE 仿真结果

$$BW_{3dB} = 1/[2\pi(R_D \parallel r_o)] = 15.9 \times 10^6 \text{ Hz}$$

可见仿真结果和计算结果一致。

瞬态分析的 HSPICE 仿真结果如图 3-21 所示。可以看出,当输入施加振幅为 0.01V 的正弦信号时,输出振幅接近 0.0194V,可见增益接近 1.94,并且输出信号与输入信号的关系是反相的。这与前述的直流、交流仿真结果及计算结果是一致的。




图 3-21 输入节点和输出节点的瞬态分析 HSPICE 仿真结果

3.4 HSPICE 电路分析进阶

直流仿真(包括工作点和直流扫描)、交流小信号仿真以及瞬态仿真是电路分析中最常用的基本仿真分析手段,可以得到绝大部分电路的基本电路特性,如数字逻辑门电路的电压转移特性、噪声容限、速度、功耗、时序等特性,而对于诸如放大器的模拟电路,通过这 3 种仿真手段,构造不同的仿真电路,可以得到直流工作点、输入/输出特性、小信号增益、幅频特性、相频特性、瞬态特性,进而得知相位裕度、带宽、压摆率(转换速率)等特性。下面介绍更多的 HSPICE 仿真分析手段,借助这些分析手段可以更详细地掌握电路的更多特性细节。

3.4.1 噪声仿真分析

在 HSPICE 中进行噪声仿真,采用 .NOISE 和 .AC 语句控制电路噪声分析。噪声分析将产生电路中每个元器件每个频率点的噪声贡献。

针对图 3-5 放大器进行噪声仿真。器件模型采用某工艺厂家提供的 $0.25\mu\text{m}$ CMOS 工艺模型文件(mix025.lib)。在进行噪声仿真之前,应使放大器处于正确的偏置下,根据前述的直流扫描结果,采用此 $0.25\mu\text{m}$ CMOS 工艺器件的放大器在输入为 $0.6\sim 0.8\text{V}$ 时是高增益区,这里设置输入为 0.75V 。采用 $0.25\mu\text{m}$ CMOS 工艺的放大器电路(见图 3-5)的噪声仿真分析的 SPICE 描述

如下。

```
* Noise analysis for AMP

* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd nch w=5u l=1.0u
M2 out 3 vdd vdd pch w=5u l=1.0u
M3 3 3 vdd vdd pch w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 0.75 AC 1.0

.OP
.AC DEC 20 100 100MEG
.NOISE v(2) Vin DEC 20 100 100MEG
.plot ac VDB(2) VP(2)
.probe noise onoise inoise
.OPTION list node post

.lib 'mix025.lib' TT

.end
```

进行噪声分析的语句为 .noise v(2) Vin DEC 20 100 100MEG, 输出节点为节点 2, 等效输入参考噪声的独立源为 Vin。按照每 10 倍频程 20 个点进行仿真, 起始频率为 100Hz, 结束频率为 100MHz。

输出噪声分析结果的语句为 .probe noise onoise inoise, 表示产生输出等效噪声以及输入等效噪声的结果。

采用 HSPICE 执行仿真, 噪声分析的结果包含在交流分析结果文件中。采用 Cscope 打开 .ac0 结果文件。单击 Signal Manager 结果文件信号列表中的 inoise(mag) 和 onoise(mag), 将显示等效输入参考噪声和输出参考噪声曲线。如图 3-22 所示, 在约 10kHz 处的输出噪声和等效输入噪声分别为 $17.447\mu\text{V}/\sqrt{\text{Hz}}$ 和 $182.61\text{nV}/\sqrt{\text{Hz}}$ 。另外, 在 HSPICE 仿真的 .lis 报表中或直接打印到屏幕的结果中, 可以找到 HSPICE 噪声分析起止频率区间内的输出噪声和等效输入噪声的总积分噪声值, 如图 3-23 所示。

3.4.2 零极点仿真分析

在 HSPICE 中采用 .PZ 语句进行零极点仿真。可以不用说明 .OP 语句, 这是由于 HSPICE 在进行零极点分析之前会自动进行工作点的分析。HSPICE 中的 .PZ 语句格式为

```
.PZ ov srcname
```




图 3-22 HSPICE 噪声仿真结果

```

**** the results of the sqrt of integral (v**2 / freq)
      using more points from fstart to fstop
      results in more accurate total noise values.

**** total output noise voltage = 4.4199m      volts
**** total equivalent input noise = 121.3637u
*****

```

图 3-23 噪声仿真报表结果

其中,srcname 是独立电压源或电流源名称;ov 可以是节点电压或支路电流。图 3-5 电路的零极点仿真分析的 SPICE 描述如下。

```

* PZ analysis for AMP
*
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

```

```

Vdd 4 0 DC 5.0
Vin 1 0 DC 1.07 AC 1.0

.OP
.PZ v(2) Vin
.OPTION list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

进行零极点分析的语句为 .pz v(2)Vin, 进行从输入信号电压源 Vin 到输出节点 2 之间的传递函数的零极点分析。

采用 HSPICE 执行仿真, 在 HSPICE 仿真的 .lis 报表中或直接打印到屏幕上的结果中, 可以找到零极点仿真分析结果的报表, 如图 3-24 所示。可见存在一个主极点, 在 -656.0246krad/s (-104.4096kHz)。这和交流仿真中得到的 3dB 带宽频率为 104kHz 的结果是相符合的。在非常远处 -65.4557Grad/s (-10.4176GHz) 还有一对零极点, 并且在传递函数中互相抵消。这是因为 HSPICE 是分别计算传递函数中的分子和分母, 因此会把结果报告出来。

```

***** pole/zero analysis          tnom= 25.000 temp= 25.000
*****

input = 0:vin                      output = v(2)

input = 0:vin                      output = v(2)

poles (rad/sec)                    poles ( hertz)
*****
real      imag                    real      imag
-656.0246k  0.                -104.4096k  0.
-65.4557g   0.                -10.4176g   0.
zeros (rad/sec)                    zeros ( hertz)
*****
real      imag                    real      imag
-65.4557g   0.                -10.4176g   0.

```

图 3-24 零极点仿真报表结果

3.4.3 传递函数仿真分析

在 HSPICE 中采用 .TF 语句计算工作点下的传递函数以及直流小信号输入输出电阻。同样地, 执行传递函数仿真会自动进行电路的工作点分析, 因此可以不必说明 .OP 语句。图 3-5 电路的传递函数仿真分析的 SPICE 描述如下。

```

* TF analysis for AMP
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u

```

```

M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 1.07

.OP
.DC Vin 1.0 1.12 0.01
.TF V(2) Vin
.OPTION list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

从前面的直流仿真结果可以得知,此放大器当输入处于 1.0~1.12V 时为高增益区。因此,这里直流扫描也限定在这个范围内,以便在传递函数仿真报表中可以聚焦这个输入偏置范围内传递函数的特性。执行 HSPICE 仿真后,传递函数仿真报表如图 3-25 所示,可见当输入为 1.07V 时,放大器的增益为 -67.1153,负号表示反相放大,输入电阻为 1.000e+20Ω(即认为是无穷大),输出电阻为 304.8651kΩ。

**** small-signal transfer characteristics				
vin	input resistance at vin	output resistance at v(2)	v(2)/vin	
1.0000	1.000e+20	16.5475k	-3.2432	
1.0100	1.000e+20	19.3547k	-3.9151	
1.0200	1.000e+20	24.6377k	-5.1368	
1.0300	1.000e+20	41.3402k	-8.8690	
1.0400	1.000e+20	328.3168k	-71.5269	
1.0500	1.000e+20	320.3343k	-70.0937	
1.0600	1.000e+20	312.5157k	-68.6200	
1.0700	1.000e+20	304.8651k	-67.1153	
1.0800	1.000e+20	297.3855k	-65.5885	
1.0900	1.000e+20	290.0788k	-64.0473	
1.1000	1.000e+20	28.9913k	-5.5674	
1.1100	1.000e+20	16.3586k	-2.7914	
1.1200	1.000e+20	12.5611k	-1.9786	

图 3-25 传递函数仿真报表结果

回顾一下此放大器直流工作点的仿真结果,当输入为 1.07V 时,如图 3-7 所示,可知 MOS 晶体管 M1 的 $g_m = 220.1475 \mu\text{S}$, $g_{ds} = 1.5059 \mu\text{S}$, M2 的 $g_{ds} = 1.7742 \mu\text{S}$,根据工作点分析结果,可以计算得到

$$r_{\text{out}} = r_{o1} \parallel r_{o2} = 1/(g_{ds1} + g_{ds2}) = 1/(1.5059 \mu\text{S} + 1.7742 \mu\text{S}) \approx 304.869 \text{k}\Omega$$

$$|A_v| = g_m \cdot r_{\text{out}} = 220.1475 \mu\text{S} \times 304.869 \text{k}\Omega \approx 67.116$$

可见与传递函数仿真分析的结果能很好地对应。同时,这些结果和交流分析中的增益以及带

宽等特性都能很好地对应上。传递函数仿真分析可以将不同输入偏置下的小信号增益以及输入输出电阻采用报表的形式直接打印出来,可以快速获知电路的特性。

3.4.4 灵敏度仿真分析

在 HSPICE 中采用 .SENS 语句分析电路特性相对于电路参数的敏感性。图 3-5 电路的灵敏度分析的 SPICE 描述如下。

```
* SENS analysis for AMP
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd 100K
.ends

X1 2 1 4 0 AMP
CL 2 0 5p

Vdd 4 0 DC 5.0
Vin 1 0 DC 1.07

.OP
.SENS V(2)
.SENS I(X1.M1)
.print dc V(2) I(X1.M1)
.OPTION list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end
```

在这个例子中,分析了节点 2 以及子电路 X1 中 M1 的偏置电流随电路参数的变化情况,执行 HSPICE 仿真后,灵敏度仿真报表如图 3-26 所示。其中,考查 $v(2)$ 对于输入电压在 1.07V 处的灵敏度,当输入电压变化 1 个单位,那么 $v(2)$ 就会变化 67.1153,这正好符合当偏置 $V_{in}=1.07V$ 时的直流小信号增益。

3.4.5 参数扫描仿真分析

HSPICE 提供功能强大的扫描仿真(Sweep)手段。下面初步介绍 Sweep 的基本用法。

HSPICE 允许在直流、交流及瞬态等仿真中增加扫描仿真。例如,可以对温度 TEMP 进行扫描,以便考查电路特性和温度之间的关系。在 .DC、.AC 及 .TRAN 等仿真语句后增加扫描说明,格式为

```
< SWEEP TEMP < START=> start < STOP=> stop < STEP=> incr >
```

```

***** dc sensitivity analysis          tnom= 25.000 temp= 25.000
*****

dc sensitivities of output v(2)

      element      element      element      normalized
      name         value       sensitivity  sensitivity
                        (volts/unit) (volts/percent)

1:r1      100.0000k  -114.6198u  -114.6198m
0:vdd     5.0000         3.5835     179.1738m
0:vin     1.0700         -67.1153   -718.1338m

***** dc sensitivity analysis          tnom= 25.000 temp= 25.000
*****

dc sensitivities of output i(x1.m1)

      element      element      element      normalized
      name         value       sensitivity  sensitivity
                        (amps/unit) (amps/percent)

1:r1      100.0000k  -172.6062p  -172.6062n
0:vdd     5.0000         5.3964u    269.8182n
0:vin     1.0700         119.0785u  1.2741u

```

图 3-26 灵敏度仿真报表结果

也可以对参数进行扫描分析。同样地,在 .DC、.AC 及 .TRAN 等仿真语句后增加扫描说明,格式为

```
< SWEEP var < START=> start < STOP=> stop < STEP=> incr >
```

针对图 3-5 电路,分别考查以下情况:①进行 $-40\sim 100^{\circ}\text{C}$ 的温度扫描,考查输入/输出特性以及交流小信号特性;②改变偏置中电阻 R1 的阻值,考查偏置对电路输入/输出特性的改变情况;③改变负载电容,考查瞬态特性的变化。参数扫描仿真分析的 SPICE 描述如下。

```

* sweep analysis for AMP
* AMP sub-circuit
.subckt AMP out in vdd gnd
.param rbias=100k load=5p
M1 out in gnd gnd MOSN w=5u l=1.0u
M2 out 3 vdd vdd MOSP w=5u l=1.0u
M3 3 3 vdd vdd MOSP w=5u l=1.0u
R1 3 gnd rbias
.ends

* X1 2 1 4 0 AMP {rbias=50k}
.param rbias=100k load=5p
X1 2 1 4 0 AMP
CL 2 0 load

Vdd 4 0 DC 5.0
Vin 1 0 DC 1.07 AC 1.0 sin(1.07 0.0001 100KHZ)

.OP
.DC Vin 0 5 0.1 sweep temp -40 100 20
* .DC Vin 0 5 0.1 sweep rbias 20k 200k 20k
.AC DEC 20 100 100MEG sweep temp -40 100 20
.TRAN 0.1u 30u sweep load 1p 10p 1p

```

```

* .print dc V(2)
* .plot dc V(2)
.probe DC V(2)
.OPTION list node post

* model
.MODEL MOSN NMOS VTO=0.7 KP=110U
+LAMBDA=0.04 GAMMA=0.4 PHI=0.7

.MODEL MOSP PMOS VTO=-0.7 KP=50U
+LAMBDA=0.05 GAMMA=0.6 PHI=0.8
.end

```

其中, .DC Vin 0 5 0.1 sweep temp -40 100 20 和 .AC DEC 20 100 100MEG sweep temp -40 100 20 语句分别同时进行温度扫描的直流仿真和交流小信号仿真,温度范围是 $-40\sim 100^{\circ}\text{C}$,步长是 20°C 。执行 HSPICE 仿真,随温度变化的直流仿真和交流仿真结果如图 3-27 所示。




图 3-27 随温度变化的直流仿真和交流仿真结果($-40\sim 100^{\circ}\text{C}$)

在输出结果的基础上,可以进一步测量随温度变化特性的曲线。这里,以交流仿真输出的幅频特性为例,绘制低频增益与温度之间的关系。在已经打开 .ac0 文件并且绘制出 v(2) 的幅频特性曲线的基础上,在 Cscope 中执行 Tools→Measurement 菜单命令,弹出如图 3-28 所示的界面,选中 dB(v(2)) 曲线,在 X value 文本框中输入 100,单击 Apply 按钮,便可绘制出 100Hz 低频交流小信号增益与温度之间的关系曲线。

如果需要考查电阻 R1 的阻值对电路输入输出特性的改变情况,那么, DC 语句描述如下。

```
.DC Vin 0 5 0.1 sweep rbias 20k 200k 20k
```

上述语句表示进行 $0\sim 5\text{V}$ 的直流扫描,步长为 0.1V ;同时进行变量参数 rbias 扫描,范围是




图 3-28 绘制 100Hz 低频交流小信号增益与温度之间的关系曲线

20~200k Ω ,步长为 20k Ω 。这里要注意,首先要定义变量参数 r_{bias}。执行 HSPICE 仿真,得到如图 3-29 所示的 v(2)的特性曲线。执行 Tools→Measurement 菜单命令测量特性曲线的阈值,在 Measurement 窗口中单击 Measurement 项右侧的 按钮,选择 General 中的 Threshold(At Y),然后在 Y value 文本框中输入 2.5,单击 Apply 按钮,得出如图 3-29 所示的 Thresh_2.5(v(2))结果。

如果需要考查负载电容对瞬态特性的影响,那么,TRAN 语句描述如下。

```
.TRAN 0.1u 30u sweep cloud 1p 10p 1p
```

上述语句表示进行 30 μ s 的瞬态仿真,步长为 0.1 μ s,同时负载电容 cloud 从 1pF 变化到 10pF,步长为 1pF。这里要注意,首先要定义变量参数 cloud。执行 HSPICE 仿真,打开 .tr0 文件并绘制 v(2),得到如图 3-30 所示的特性曲线。执行 Tools→Measurement 菜单命令测量特性曲线的阈值,在 Measurement 窗口中单击 Measurement 项右侧的 按钮,选择 Levels 中的 Peak to Peak,然后单击 Apply 按钮,得出如图 3-30 所示的瞬态波形的峰峰值与 cloud 之间的关系曲线 PK2PK(v(2))。

3.4.6 工艺角仿真分析

在 3.3 节例 3-1 中介绍过,实际的工艺厂家模型文件会提供不同工艺角,以便描述工艺的偏移情况。在 HSPICE 仿真中,可以采用 .ALTER 语句更换工艺中不同的工艺角,这样就可以对工艺的偏移对电路性能造成的影响进行仿真分析。以直流仿真为例,进行不同工艺角仿真分析的 SPICE 描述如下。




图 3-29 随偏置电阻变化的直流仿真结果曲线




图 3-30 随负载电容变化的瞬态仿真结果曲线

```

* DC analysis for AMP under corners
* AMP sub-circuit
.subckt AMP out in vdd gnd
M1 out in gnd gnd nch w=5u l=1.0u
M2 out 3 vdd vdd pch w=5u l=1.0u
M3 3 3 vdd vdd pch w=5u l=1.0u
R1 3 gnd 100K
    
```

```

.ends

X1 2 1 4 0 AMP
CL 2 0 5p
Vdd 4 0 DC 5.0
Vin 1 0 DC 5.0

.OP
.OPTION list node post

.lib 'mix025.lib' TT
.ALTER
.lib 'mix025.lib' SS
.ALTER
.lib 'mix025.lib' FF
.end

```

执行 HSPICE 仿真后,由于进行了 3 个工艺角的仿真,在工作目录中产生了 .sw0~.sw2 这 3 个直流扫描结果文件。用 Cscope 工具打开这些文件,如图 3-31 所示。然后在每个信号列表中双击 v(2),得到输出节点相对于 Vin 的直流扫描结果,如图 3-32 所示,可以发现输入/输出特性出现了




图 3-31 3 个直流扫描结果文件




图 3-32 3 个工艺角的直流仿真结果

变化,在不同的工艺角下高增益区间是不一样的,也就是说对于这样的放大器,在工艺发生变化时,实现放大功能需要改变输入偏置。这说明图 3-5 这样的简单共源极放大器对工艺的容忍程度较差。

3.5 本章小结

本章描述了基于 HSPICE 的集成电路仿真方法。HSPICE 是业界作为提交数据的标准电路级仿真工具之一。HSPICE 在兼容标准 SPICE 的基础上,提供更强大的集成电路仿真分析能力。同时,HSPICE 采用和标准 SPICE 描述一致的输入文件,其仿真更具有“原汁原味”SPICE 的特点。因此,学习和掌握 HSPICE 有助于深刻理解集成电路的电路级仿真分析。

HSPICE 也在一直发展更新,更多 HSPICE 功能请读者参考相关的 HSPICE 用户手册及参考文档。